

Lenka Pospíšilová

OBÁLKY ROVINNÝCH KŘIVEK S PROGRAMEM MAPLE

Abstrakt

V následujícím textu naznačíme, jakým způsobem je možné využít Maple při výuce diferenciální geometrie křivek. V příspěvku se zaměříme na téma obálek rovinných křivek. Uvedený programový kód byl odzkoušen pro verzi Maple 9.5.

Klíčová slova

rovinná křivka, obálka rovinných křivek, Maple.

1 Úvod

Kurz diferenciální geometrie křivek patří k tradičním předmětům vyučovaným na matematických katedrách. Vzhledem k neustálému vývoji výpočetní techniky a její dostupnosti na vysokých školách se nabízí jisté propojení klasické výuky s výukou v počítačové učebně. Je však třeba důsledně dbát, aby studenti pro každý výpočet provedený s pomocí počítače zvážili jeho správnost. Zavedení výpočetní techniky do výuky diferenciální geometrie navíc nemá nahradit schopnost studentů vypočítat příklad metodou "tužka - papír". Význam počítačové podpory výuky spočívá v usnadnění technicky zdlouhavých výpočtů a tím také k větším možnostem experimentování. Dále má nepostradatelný význam z hlediska grafického znázornění, které může pomoci lepšímu pochopení problematiky. Nutnost hledání algoritmizace při využívání matematického softwaru pak rozvíjí určitý druh logického myšlení.

2 Obálka rovinných křivek

Způsob řešení příkladů na výpočet obálky rovinných křivek během výuky na počítačové učebně naznačíme při řešení dvou úloh. První úloha vychází ze soustavy rovinných křivek, která je daná rovnicí $F(x, y, c) = 0$. Tento typ úlohy je vzhledem k rozličnému průběhu

výpočtu přijatelnější řešit postupnými příkazy. Jinak je třeba si uvědomit, že rovnice soustavy křivek je často v zadání příkladu skryta. Studenti musí použít vlastní úvahu, aby rovnici soustavy vyjádřili. Tohle je ve výuce důležitý moment, který poukazuje na nutnost neustále aktivního přístupu při řešení úloh s počítačem.

Druhá úloha naopak poukazuje na situaci, kdy se jistá zobecnění pomocí uživatelské procedury přímo nabízí. Tento druh abstrakce způsobuje studentům problémy, přesto je přínosné, aby se s ním seznámili a naučili se ho používat.

2.1 Obálka soustavy rovinných křivek zadané rovnicí $F(x, y, c) = 0$.

Uvažujme například obálku soustavy kružnic

$$F(x, y, t) = (x - t)^2 + y^2 - \frac{t^2}{4} = 0,$$

přičemž $t \in (-\infty, 0) \cup (0, \infty)$. Nadefinujeme si nejdříve rovnici soustavy kružnic, její parciální derivaci podle parametru soustavy a charakteristickou množinu:

```
> restart:with(plots):
> F:=(x-t)^2+y^2-t^2/4=0:
> dF:=diff(F,t):
> sol:=allvalues(solve({F,dF},{x,y}));
```

$$sol := \left\{ x = \frac{3t}{4}, y = \frac{\sqrt{3}t}{4} \right\}, \left\{ x = \frac{3t}{4}, y = -\frac{\sqrt{3}t}{4} \right\}$$


Máme dvě řešení, která přiřadíme pomocí příkazu `assign` do proměnných `e1` a `e2`. Vždy je následně nutné použít příkaz `unassign` pro uvolnění přiřazených proměnných! Vykreslení obálky zahrnuje proměnná `ENV`:

```
> assign(sol[1]):
> env1:=unapply([x,y],t):unassign('x','y'):
> assign(sol[2]):
> env2:=unapply([x,y],t):unassign('x','y'):
> ENV:=plot([env1(t)[1],env1(t)[2],t=-9..9],
>           [env2(t)[1],env2(t)[2],t=-9..9]),
>           thickness=3,color=red):
```

OBÁLKY ROVINNÝCH KŘÍVEK S PROGRAMEM MAPLE

Vykreslíme obálku společně s několika kružnicemi zadané soustavy. V Maplu je příhodnější zobrazovat křivky určené parametrickými rovnicemi a ne implicitně. Proto v jednodušších případech převedeme implicitní zadání na parametrické a pak teprve objekt vykreslíme. Soustava kružnic parametricky:

```
> f:=(s,t)->[t+t/2*cos(s),t/2*sin(s)]:  
> SYS:=seq(plot([f(s,t/6)[1],f(s,t/6)[2],s=0..2*Pi],  
> color=grey),t=-30..30):  
> display({SYS,ENV},scaling=constrained);
```



2.2 Obálka normál rovinné křivky

Obálka normál (evoluta) rovinné křivky odpovídá množině středů oskulačních kružnic. Hledání těchto středů se v Maplu implementuje snažším způsobem nežli výpočet obálky normál. Navíc umožňuje demonstrovat studentům tvorbu a použití uživatelské procedury. Pro střed oskulační kružnice křivky dané parametrizací

$$f(t) = (f_1(t), f_2(t))$$

v bodě $f(t_0)$ platí:

$$x = f_1(t_0) - \frac{f_2'(t_0)(f_1'(t_0)^2 + f_2'(t_0)^2)}{f_1'(t_0)f_2''(t_0) - f_2'(t_0)f_1''(t_0)}$$
$$y = f_2(t_0) + \frac{f_1'(t_0)(f_1'(t_0)^2 + f_2'(t_0)^2)}{f_1'(t_0)f_2''(t_0) - f_2'(t_0)f_1''(t_0)}$$

Vytvoříme proceduru `evolute`, jejímž vstupním parametrem bude křivka. Lokální proměnná `r_osc` reprezentuje spojnicí středu oskulační kružnice s bodem dotyku, lokální proměnné `df`, `ddf`, `dfdf`, `dfddf` představují po řadě $f'(t)$, $f''(t)$, $f_1'^2 + f_2'^2$, $\begin{vmatrix} f_1' & f_2' \\ f_1'' & f_2'' \end{vmatrix}$:

```
> restart:with(plots):
```

Lenka Pospíšilová

```
> evolute:=proc(f)
>   local df,ddf,dfdf,dfddf,r_osc;
>   df:=diff(f(t),t);
>   ddf:=diff(f(t),t$2);
>   dfdf:=df[1]*df[1]+df[2]*df[2];
>   dfddf:=df[1]*ddf[2]-df[2]*ddf[1];
>   r_osc:=t->[-df[2]*dfdf/dfddf,df[1]*dfdf/dfddf];
>   simplify(evalm(f(t)+r_osc(t)));
> end:
```

Evoluta cykloidy:

```
> cycloid:=t->[r*(t-sin(t)),r*(1-cos(t))]:
> evolute(cycloid);
```

$$[(\sin(t) + t)r, r(-1 + \cos(t))]$$


Následující kód implementuje proceduru `plot_evolute` zobrazující normály křivky. Obálkou těchto normál je hledaná evoluta, která je patrná i bez jejího vlastního znázornění. Procedura má za vstupní parametry křivku `f`, interval vykreslení (t_1, t_2) a číslo `k` udávající počet vykreslovaných normál `N_PLOT`:

```
> plot_evolute:=proc(f,t1,t2,k)
>   local df,ddf,dfdf,dfddf,r_osc,n_par,N_PLOT,F_PLOT;
>   df:=diff(f(t),t);
>   ddf:=diff(f(t),t$2);
>   dfdf:=df[1]*df[1]+df[2]*df[2];
>   dfddf:=df[1]*ddf[2]-df[2]*ddf[1];
>   r_osc:=t->[-df[2]*dfdf/dfddf,df[1]*dfdf/dfddf];
>   n_par:=evalm(f(t)+s*r_osc(t));
>   N_PLOT:=seq(plot([n_par[1],n_par[2],s=-1..1],
>                   color=blue),t=seq(t1+(t2-t1)*i/k,i=0..k)):
>   F_PLOT:=plot([f(t)[1],f(t)[2],t=t1..t2],
>               thickness=3,color=red):
>   display({N_PLOT,F_PLOT},scaling=constrained);
> end:
```

Obálka normál cykloidy:

```
> r:=1:
> plot_evolute(cycloid,0.1,8*Pi,60);
```

OBÁLKY ROVINNÝCH KŘÍVEK S PROGRAMEM MAPLE




Obálka normál elipsy:

```
> ellipse:=t->[a*cos(t),b*sin(t)];
```


```
> evolute(ellipse);
```

$$\left[\frac{\cos(t)^3(-b^2 + a^2)}{a}, -\frac{\sin(t)^3(-b^2 + a^2)}{b} \right]$$

```
> a:=5:b:=3:plot_evolute(ellipse,0,2*Pi,32);
```




Uvedené procedury můžeme vyzkoušet např. také při hledání obálky normál paraboly, traktrixu, srdcovky nebo logaritmické spirály:



Obdobným způsobem jako při vykreslování obálky normál můžeme vykreslit obálku tečen. Získané obrázky dokládají studentům tvrzení uváděné na přednáškách, že křivka bez inflexních bodů je obálkou svých tečen.

Soustava tečen pro elipsu, parabolu, traktrix, srdcovku a logaritmickou spirálu:



3 Závěr

Problematika obálek rovinných křivek posloužila k nástinu výuky, která proběhla v akademickém roce 2004/2005 na Katedře matematiky PřF MU v Brně v rámci kursů diferenciální geometrie křivek a diferenciální geometrie ploch. Cvičení k těmto předmětům probíhala na počítačové učebně v rozsahu 2 hodin týdně. V rámci těchto cvičení byl zachován klasický styl výuky a poté byl doplněn výukou ve spolupráci s programem Maple. Tímto způsobem vedená výuka měla od studentů pozitivní odezvu, při které jsem se setkala se dvěma opačnými jevy; kdy výpočetní technika přiblížila studentům svět diferenciální geometrie a stejně tak počítačová řešení příkladů v diferenciální geometrii přiblížila některým studentům svět výpočetní techniky.

Literatura

- [1] K. Rektorys: *Přehled užití matematiky I.*, Prometheus, Praha, 2003
- [2] V. Rovenski: *Geometry of Curves and Surfaces with MAPLE*, Birkhäuser, Boston, 2000